MS 365 Excel Basics #2

Adding & Counting with Functions like SUMIFS, COUNTIFS, COUNT, ROWS and IF

Table of Contents

COUNT, COUNTA, ROWS, SUM, SUMIFS, COUNTIFS & IF functions

- **COUNT** function: Counts numbers in an array or range of cells and ignores text, logicals, errors and empty cells.
  - COUNT(value1, [value2], ...)
    - value1 Required. The first item, cell reference, or range within which you want to count numbers.
    - value2, ... Optional. Up to 255 additional items, cell references, or ranges within which you want to count numbers.
- **COUNTA** function: Count all the cells that are not empty in an array or range of cells.
  - COUNTA(value1, [value2], ...)
    - value1 Required. The first argument representing the values that you want to count.
    - value2, ... Optional. Additional arguments representing the values that you want to count, up to a maximum of 255 arguments.
- **ROWS** function: Counts number of rows in an array or range of cells.
  - ROWS(array)
    - Array Required. An array, an array formula, or a reference to a range of cells for which you want the number of rows.
    - Example: ROWS(B3:B7) = 5 because the range spans the rows: 3, 4, 5, 6, 7.
- **SUM** function adds numbers from an array or range of cells.
  - SUM(number1, [number2], ...) = adds numbers to get a total.
    - number1 Required. The first number, cell reference, or range for which you want the average.
    - number2, ...Optional. Additional numbers, cell references or ranges for which you want the average, up to a maximum of 255.
- **COUNTIFS** function: Counts just some of the items in a range of cells based on a condition of set of criteria. Can count with 1 or more conditions/criteria.
  - COUNTIFS(criteria_range1, criteria1, [criteria_range2, criteria2]...)
    - criteria_range1 The first range with the items that you want to match to the value you enter into criteria1
    - criteria1 argument contains the condition or criteria that tells the function what to count. Conditions and criteria can be numbers, text, logical, such as 10/23/2025, 500, >500, Luong.
    - [criteria_range2, criteria2]... Optional pairs of range and criteria that allow you to count based on an AND Logical Test. Up to 127 pairs of ranges and criteria.
- **SUMIFS** function: Add just some of the numbers in a range based on a condition of set of criteria. Can add with 1 or more conditions/criteria.
  - SUMIFS(sum_range, criteria_range1, criteria1, [criteria_range2, criteria2], ...)
    - sum_range ... Contains the range with the numbers that you want to add.
    - criteria_range1 The first range with the items that you want to match to the value you enter into criteria1
    - criteria1 argument contains the condition or criteria that tells the function what to count. Conditions and criteria can be numbers, text, logical, such as 10/23/2025, 500, >500, Luong.
    - [criteria_range2, criteria2]... Optional pairs of range and criteria that allow you to count based on an AND Logical Test. Up to 127 pairs of ranges and criteria.
- **IF** function: Returns a value based on a TRUE or FALSE from a logical test. The IF function delivers one value when the logical test evaluates to TRUE and a different value when the logical test evaluates to FALSE.
  - =IF(logical_test, [value_if_true], [value_if_false])
    - logical_test Contains a single logical test or an array logical test.
    - value_if_true This is the value that is delivered if the logical test evaluates to TRUE.
    - value_if_false This is the value that is delivered if the logical test evaluates to FALSE.

UNIQUE array function

The UNIQUE array function can deliver either a distinct set of items or a unique set of items from a table, a column, or a row:

- A *distinct* set of items has all duplicates in the set removed, leaving only one of each item.
- A *unique* set of items lists only items that occur exactly one time in the dataset.

Note: In everyday language, when people say they want a unique list, they are asking for a set of items that has all duplicates in the set removed, leaving only one of each item. Microsoft chose the term distinct to represent this action, rather than unique.

The UNIQUE function has the following syntax (see Figure 3.18):

```
UNIQUE(array, [by_col], [exactly_once])
```

The arguments of the UNIQUE function are as follows:

- ***array*:** This argument contains the table, column, or row that you want to filter:
  - When you provide a table, the function generates a unique/distinct set of records.
  - When you provide a column, the function generates a unique/distinct list of items in a column.
  - When you provide a row, the function generates a unique/distinct list of items in a row.
- **[*by_col*]:** This argument can take one of two possible values:
  - TRUE or 1: Returns all unique/distinct columns from the array.
  - FALSE or omitted or 0: Returns all unique/distinct rows from the array. This the default.
- **[*exactly_once*]:** This argument can take one of two possible values:
  - TRUE or 1: Returns a unique set as defined above.
  - FALSE or 0 or omitted: Returns a distinct set as defined above.



Figure 3.18 *Arguments of the UNIQUE array function.*

SORT array function

The SORT array function can sort a row, a column, or a table in ascending or descending order. It can also sort by more than one column. The SORT function has the following syntax (see Figure 3.24):

```
SORT(array, [sort_index], [sort_order], [by_col])
```

The arguments of the SORT function are as follows:

- ***array*:** This argument contains the table, column, or row that is the source data.
- **[*sort_index*]:** You use this argument when you have more than two columns or rows and you want to specify which column or row should be used for sorting. This argument indicates the relative position of the column or columns that you want to sort by. When this argument is omitted, if you are sorting a table by rows, the first column in the table is used to sort the records; if you are sorting a table by columns, the first row in the table is used to sort the records. Consider these examples:
  - Sort by column 2: =SORT(D5:E10,2)
  - Sort by column 1 and then by column 3: =SORT(C5:E10,{1,3})
- **[*sort_order*]:** This argument can take the following values:
  - 1 or omitted: A–Z, or smallest-to-largest, sort. This is the default.
  - -1: Z–A, or largest-to-smallest, sort.
  - If you want to sort two or more columns, each with a different sort, use array syntax, such as =SORT(C5:E10,{3,1},{1,-1}), where column 3 gets an A–Z sort and column 1 gets a Z–A sort.
- **[*by_col*]:** This argument can take one of two values:
  - TRUE: Sort by columns. Use this option when you are sorting a row of values.
  - FALSE or omitted: Sort by rows. Use this option when you are sorting a column. This is the default.



Figure 3.24 *Arguments in the SORT array function.*

Examples on Next Page:

Date	SalesRep	Sales ($)
10/23/2025	Luong	100
10/23/2025	Sioux	200
10/24/2025	Chantel	100
10/24/2025	Luong	300
10/24/2025	Luong	
10/23/2025	Chantel	100
10/24/2025	Sioux	200
10/24/2025	Sioux	500
10/23/2025	Sioux	200
10/23/2025	Luong	500
10/23/2025	Sioux	300
10/24/2025	Chantel	500
10/24/2025	Chantel	700
10/24/2025	Luong	800
10/23/2025	Chantel	400
10/24/2025	Sioux	400
10/24/2025	Chantel	500
10/23/2025	Luong	400

COUNT	COUNTA	ROWS	SUM
Counts Numbers	Counts cells not empty	Counts rows in a range	Adds numbers
Count ALL Numbers	Count ALL Words	Counts ALL Rows	Sum ALL Numbers
17	18	18	6200
=COUNT(D3:D20)	=COUNTA(C3:C20)	=ROWS(D3:D20)	=SUM(D3:D20)

****When you specify a "criteria" or "condition" you are saying: "don't make the calculation on all the items, just on some of the items".**

	COUNTIFS	SUMIFS
	Count w/ 1 or more criteria	Add w/ 1 or more criteria
Criteria	Count number of sales that Chantel made	Adds the sales that Chantel made
Chantel	6	2300
	=COUNTIFS(C3:C20,F14)	=SUMIFS(D3:D20,C3:C20,F14)

Criteria	Count the number of sales made on 10/24/2025	Add sales made on 10/24/2025
10/24/2025	10	4000
	=COUNTIFS(B3:B20,F18)	=SUMIFS(D3:D20,B3:B20,F18)

Goal: Create SalesRep Sales Report:

SORT	
Sorts A-Z or Z-A	
UNIQUE	SUM
Creates a unique list	Adds numbers
SalesRep	Sales ($)
Chantel	2,300
Luong	2,100
Sioux	1,800

F28: =UNIQUE(SORT(C3:C20))

G28: =SUMIFS(D3:D20,C3:C20,F28#)

Excel's Golden Rule

- If a formula input can change, put it in a cell, label it, and refer to it in the formula with a cell reference. If the input will never change, like 24 hours in a day, then you can hard code it into formula.
    - Examples of formula inputs that can change: SalesRep name like: Luong, Sales amount like: 100, Tax Rates like: 0.0375 or 3.75%, Sales Hurdles like: >500.
    - Examples of formula inputs that will not change: Months in a year = 12, Hours in a day = 24, Days in a week = 7.

Before:

Date	SalesRep	Sales ($)
10/23/2025	Luong	100
10/23/2025	Sioux	200
10/24/2025	Chantel	100
10/24/2025	Luong	300
10/24/2025	Luong	
10/23/2025	Chantel	100
10/24/2025	Sioux	200
10/24/2025	Sioux	500
10/23/2025	Sioux	200
10/23/2025	Luong	500
10/23/2025	Sioux	300
10/24/2025	Chantel	500
10/24/2025	Chantel	700
10/24/2025	Luong	800
10/23/2025	Chantel	400
10/24/2025	Sioux	400
10/24/2025	Chantel	500
10/23/2025	Luong	400

Goal:
1) Create an Excel solution that counts how many sales Chantel made.
2) Add sales for the day 10/24/2025.

Goal:
If the insurances expanse for the year is $3,000, and the company records the same amount for each month, what is the monthly insurance expense?

After completing the problem and following Excel's Golden Rule:

Date	SalesRep	Sales ($)
10/23/2025	Luong	100
10/23/2025	Sioux	200
10/24/2025	Chantel	100
10/24/2025	Luong	300
10/24/2025	Luong	
10/23/2025	Chantel	100
10/24/2025	Sioux	200
10/24/2025	Sioux	500
10/23/2025	Sioux	200
10/23/2025	Luong	500
10/23/2025	Sioux	300
10/24/2025	Chantel	500
10/24/2025	Chantel	700
10/24/2025	Luong	800
10/23/2025	Chantel	400
10/24/2025	Sioux	400
10/24/2025	Chantel	500
10/23/2025	Luong	400

Goal:
1) Create an Excel solution that counts how many sales Chantel made.
2) Add sales for the day 10/24/2025.

SalesRep	Chantel
Number sales Chantel made	6

=COUNTIFS(C3:C20,G7)

Date	10/24/2025
Total sales on 10/24/2025	4000

=SUMIFS(D3:D20,B3:B20,G10)

Goal:
If the insurances expanse for the year is $3,000, and the company records the same amount for each month, what is the monthly insurance expense?

Year Insurance Expense	3,000.00
Monthly Insurance Expense	250.00

=G18/12

$ Sign Number Formatting

- General Number Formatting:
  - General Number Formatting = What you see is what is in the cell.
  - General Number Formatting ERASES all previously applied Number Formatting.
- Accounting Number Format:
  - Fixed dollar sign (left edge of cell).
  - Negatives are in parenthesis.
  - Zeros are dashes.
  - Decimals always line up.
- Currency:
  - Floating dollar sign.
  - You choose how to show negatives.
  - Zeros are zeros.
  - Decimals usually line up.
- Example:

General
Sales
45
0
78.99
100
-101
98.2
20

Currency
Sales
$45.00
$0.00
$78.99
$100.00
-$101.00
$98.20
$20.00

Accounting	
Sales	
$	45.00
$	-
$	78.99
$	100.00
$	(101.00)
$	98.20
$	20.00

SEQUENCE Array Function

- The SEQUENCE Array Function generates a sequence of numbers in a row, a column, or a table, based on a start value and an increment value (step).
  - SEQUENCE(rows, [columns], [start], [step])
    - Rows requires the number of rows you want in the sequence of numbers.
    - [columns] requires the number of columns you want in the sequence of numbers.
    - [start] requires the start number for the the sequence of numbers.
    - [step] requires the step, or increment between each number in the sequence of numbers

AND Logical Test

Single condition logical test: Mom says: "If you take out the garbage, you get dessert".
If "take out garbage" = TRUE, you get dessert

Two condition logical test: Mom says: "If you take out the garbage AND clean the table, you get dessert".
If "take out garbage" = TRUE AND "clean the table" = TRUE, you get dessert
TRUE, TRUE = you get dessert

AND Logical Test =
Two or more logical tests are used to test whether to count or add an item. All tests must be met, for the item to be included.

Four possibilities for an AND Logical Test: "If you take out the garbage AND clean the table, you get dessert".
TRUE, TRUE = TRUE = you get dessert
TRUE, FALSE = FALSE = No dessert
FALSE, TRUE = FALSE = No dessert
FALSE, FALSE = FALSE = No dessert

Goal: Count sales on 10/23/2025 made by Luong

The AND Logical Test is:
the Date Field must contain 10/23/2025
AND
the SalesRep Field must contain a number that is Luong.

Criteria:

Date	SalesRep
10/23/2025	Luong

In Order to Count The Record, you Must Get Two TRUEs!!!!

Date	SalesRep	Sales ($)
10/23/2025	Sioux	200
10/23/2025	Chantel	100
10/24/2025	Luong	300
10/23/2025	Luong	100

AND Logical Test must get Two TRUEs:

Date	SalesRep	Sales ($)
TRUE	FALSE	X
TRUE	FALSE	X
FALSE	TRUE	X
TRUE	TRUE	✓

Count sales on 10/23/2025 made by Luong = 1

SUMIFS & COUNTIFS with 2 Conditions in an AND Logical Test

Date	SalesRep	Sales ($)
10/23/2025	Luong	100
10/23/2025	Sioux	200
10/26/2025	Bree	900
10/24/2025	Chantel	100
10/24/2025	Luong	300
10/24/2025	Luong	
10/23/2025	Chantel	100
10/26/2025	Bree	800
10/24/2025	Sioux	200
10/24/2025	Sioux	500
10/23/2025	Sioux	200
10/23/2025	Luong	500
10/23/2025	Sioux	300
10/24/2025	Chantel	500
10/24/2025	Chantel	700
10/24/2025	Luong	800
10/23/2025	Chantel	400
10/24/2025	Sioux	400
10/26/2025	Bree	700
10/24/2025	Chantel	500
10/23/2025	Luong	400

SUMIFS and COUNTIFS (other ...IFS too) perform AND Logical Tests by defult

Example: Count sales on 10/23/2025 made by Luong
AND Logical Test is: **Date = 10/23/2025 AND SalesRep = Luong**

Date	10/23/2025
SalesRep	Luong
Count with 2 criteria	3
Sum with 2 criteria	1000

G51: =COUNTIFS(B45:B65,G49,C45:C65,G50)
G52: =SUMIFS(D45:D65,B45:B65,G49,C45:C65,G50)

Comparative Operators.

=	Equal: are two things equal?
<>	Not: are two things not equal? Type less than symbol, then greater than symbol.
>	Greater than: is the thing on the left greater than the thing on the right?
>=	Greater than or equal to: is the thing on the left greater than or equal to the thing on the right?
<	Less than: is the thing on the left less than the thing on the right?
<=	Less than or equal to: is the thing on the left less than or equal to the thing on the right?

Example: Count Sales on 10/24/2025 that were greater than or equal to 500
AND Logical Test is: **Date = 10/24/2025 AND Sales >=500**

Date	10/24/2025
Sales	>=500
Count with 2 criteria	5
Sum with 2 criteria	3000

G66: =COUNTIFS(B45:B65,G64,D45:D65,G65)
G67: =SUMIFS(D45:D65,B45:B65,G64,D45:D65,G65)

Comparative Operators

Comparative Operators:

=	Equal: are two things equal?
<>	Not Equal: are two things not equal. Type less than symbol, then greater than symbol
>	Greater than: is the thing on the left greater than the thing on the right?
>=	Greater than or equal to: is the thing on the left greater than or equal to the thing on the right?
<	Less than: is the thing on the left less than the thing on the right?
<=	Less than or equal to: is the thing on the left less than or equal to the thing on the right?

There are three situations where we can use comparative operators in worksheet formulas:

Date	SalesRep	Sales ($)
10/23/25	Luong	100
10/23/25	Sioux	200
10/26/25	Bree	900
10/24/25	Chantel	100
10/24/25	Luong	300
10/24/25	Luong	
10/23/25	Chantel	100
10/26/25	Bree	800
10/24/25	Sioux	200
10/24/25	Sioux	500
10/23/25	Sioux	200
10/23/25	Luong	500
10/23/25	Sioux	300
10/24/25	Chantel	500
10/24/25	Chantel	700
10/24/25	Luong	800
10/23/25	Chantel	400
10/24/25	Sioux	400
10/26/25	Bree	700
10/24/25	Chantel	500
10/23/25	Luong	400

1) COUNTIFS, SUMIFS and other IFS function options #1: Type >=500 in cell

You can type the comparative operator and the number into the cell.
The comparative operator and number are considered a text value, which works in COUNTIFS and other IFS functions.
Microsoft programmed the COUNTIFS and other IFS functions to understand a comparative operator and number as text values.
Use this method when the hurdle number will not be used as a number in other formulas

Example: Count Sales on 10/24/2025 that were greater than or equal to 500
AND Logical Test is: **Date = 10/24/2025 AND Sales >=500**

Date	10/24/2025
Sales Hurdle	>=500
Count with 2 criteria	5

=COUNTIFS(D13:D33,G22,B13:B33,G21)

For IFS functions: you can type the comparative operator and the number into the cell as a text value.

2) COUNTIFS, SUMIFS and other IFS function options #2: Join text comparative operator with number in criteria argument

When the hurdle number is used as a number in other formulas, then you cannot type the comparative operator and number into a cell as a text value.
When this is the case, you can use the ampersand to join a comparative operator in quotes to the cell with the number in the criteria argument of the IFS function, as shown below:

Example: Count Sales on 10/24/2025 that were greater than 500
AND Logical Test is: **Date = 10/24/2025 AND Sales 500**

Date	10/24/2025
Count Sales Above 500	500
Count with 2 criteria	2

F33: ="Count Sales Above "&G33
G34: =COUNTIFS(D13:D33,">"&G33,B13:B33,G32)

For IFS functions: you can Join text comparative operator with number in criteria argument

Credit Rating	Logical Test	Single-Input / Single Output
4.6	TRUE	TRUE
3.7	FALSE	FALSE
4	TRUE	FALSE
2	FALSE	FALSE
1.6	FALSE	FALSE
2.9	FALSE	FALSE

C37: =B37:B42>=G44
D37: =B37>$G$44

3) Formula Logical Tests:

When creating a Logical Formula, such as an Array Logical Test:
You type the range of cells or cell on the left, then the comparative operator, then the range of cells or cell on the right.
You do not type the comparative operator into the formula as text.

Example: Is the Credit Rating greater then or equal to 4?
Single Condition Logical Test: **Credit Rating >= 4**

Credit Rating >=	4

For Logical Formulas that will deliver TRUE and FALSE values, type comparative operator directly into formula (not text)

SUM and SUMIFS & COUNTIFS with 2 Conditions in an OR Logical Test Over One Column

Two condition logical test: Mom says: "If you take out the garbage OR clean the table, you get dessert".

You get dessert if you just take out garbage, or you just clean the table, or you do both!

OR Logical Test

One or more tests needs to come out true in order to count or add an item.

You must get at leaset one TRUE, for the item to be included.

For two tests in an OR logical test, these are the four possibilities:

TRUE, TRUE = TRUE 2 TRUEs

FALSE, TRUE = TRUE 1 TRUE

TRUE, FALSE = TRUE 1 TRUE

FALSE, FALSE = FALSE 0 TRUE

Date	SalesRep	Sales ($)
10/23/2025	Luong	100
10/23/2025	Sioux	200
10/26/2025	Bree	900
10/24/2025	Chantel	100
10/24/2025	Luong	300
10/24/2025	Luong	
10/23/2025	Chantel	100
10/26/2025	Bree	800
10/24/2025	Sioux	200
10/24/2025	Sioux	500
10/23/2025	Sioux	200
10/23/2025	Luong	500
10/23/2025	Sioux	300
10/24/2025	Chantel	500
10/24/2025	Chantel	700
10/24/2025	Luong	800
10/23/2025	Chantel	400
10/24/2025	Sioux	400
10/26/2025	Bree	700
10/24/2025	Chantel	500
10/23/2025	Luong	400

SUMIFS and COUNTIFS (other ...IFS too) DO NOT perform OR Logical Tests by default

For an OR Logical Test on **one column**, use: SUM(COUNTIFS()) and SUM(SUMIFS())

Example: Count and add sales for teach Sales Team

ST1 logical test for count: **SalesRep = Luong OR SalesRep = Sioux**

SalesTeam 1	SalesTeam 2
Luong	Bree
Sioux	Chantel

	Sales Team 1	Sales Team 2
Count Sales	12	9
Add Sales	3900	4700

=SUM(COUNTIFS($C$83:$C$103,G89:G90))

=SUM(SUMIFS($D$83:$D$103,$C$83:$C$103,G89:G90))

MAX Sales	800	900
MIN Sales	100	100
Average Sales	325	522.2222222

=MAX(MAXIFS($D$83:$D$103,$C$83:$C$103,G89:G90))

=MIN(MINIFS($D$83:$D$103,$C$83:$C$103,G89:G90))

=AVERAGE(IF(($C$83:$C$103=G89)+($C$83:$C$103=G90),$D$83:$D$103))

Array Logical Test with SUM & IF Functions to Run an OR Logical Test Over Two Different Columns

Last Year Sales	Credit Rating	Customer
1,250,000	4.6	SW
955,500	3.7	PCC
875,000	4	QFC
2,100,500	2	FM
550,750	1.6	WM
2,500,000	2.9	L

Last Year Sales	Credit Rating	Add
TRUE	TRUE	2
FALSE	FALSE	0
FALSE	TRUE	1
TRUE	FALSE	1
FALSE	FALSE	0
TRUE	FALSE	1

Example: Our company only extends credit to customers who have:

OR Logical Test is: **Last Year Sales > 1,000,000 OR Credit Rating >=4.**

Last Year Sales >	1,000,000
Credit Rating >=	4
Count Customers	4
Add Sales	6,725,500

=SUM(IF((B107:B112>G109)+(C107:C112>=G110),1,0))

=SUM(IF((B107:B112>G109)+(C107:C112>=G110),B107:B112,0))

We CANNOT use SUMIFS and COUNTIFS

(or any other "IFS" functions) for an OR Logical Test across two columns!!!!!

For an OR Logical Test on two or more columns, use:

SUM and **IF** functions together with an Array **Logical Test**.

For an **Array Logical Test** in Excel:

1) AND Logical Test uses * math operator
2) OR Logical Test uses + math operator
3) Any math operator used on TRUE and FALSE converts TRUE to 1 and FALSE to 0
4) Excel interprets any non-zero number as TRUE, and zero as FALSE.

AND Logical Test: Use * multiplication

TRUE * TRUE = 1 * 1 = 1
TRUE * FALSE = 1 * 0 = 0
FALSE * TRUE = 0 * 1 = 0
FALSE * FALSE = 0 * 0 = 0

OR Logical Test: Use + Addition

TRUE + TRUE = 1 + 1 = 2
TRUE + FALSE = 1 + 0 = 1
FALSE + TRUE = 0 + 1 = 1
FALSE + FALSE = 0 + 0 = 0

Alternative Formulas for an OR Logical Test Over Two Columns:

Other Methods for OR Logical Tests Across Two Columns:

Last Year Sales >	1,000,000	
Credit Rating >=	4	
Count Customers	4	=SUM(IF((B107:B112>C142)+(C107:C112>=C143),1))
Count Customers	4	=COUNTIFS(B107:B112,">"&C142)+COUNTIFS(C107:C112,">="&C143)-COUNTIFS(B107:B112,">"&C142,C107:C112,">="&C143)
Count Customers	4	=ROWS(FILTER(D107:D112,(B107:B112>C142)+(C107:C112>=C143)))
Count Customers	4	=DCOUNTA(B106:D112,D106,B153:C155)
Add Sales	6,725,500	=SUM(IF((B107:B112>C142)+(C107:C112>=C143),B107:B112))
Add Sales	6,725,500	=SUMIFS(B107:B112,B107:B112,">"&C142)+SUMIFS(B107:B112,C107:C112,">="&C143)-SUMIFS(B107:B112,B107:B112,">"&C142,C107:C112,">="&C143)
Add Sales	6,725,500	=SUM(FILTER(B107:B112,(B107:B112>C142)+(C107:C112>=C143)))
Add Sales	6,725,500	=DSUM(B106:D112,B106,B153:C155)

Last Year Sales	Credit Rating
>1000000	
	>=4

Other Aggregate Calculations for OR Logical Tests Across Two Columns:

MAX Sales	2,500,000	=MAX(IF((B107:B112>C142)+(C107:C112>=C143),B107:B112))
MIN Sales	875,000	=MIN(IF((B107:B112>C142)+(C107:C112>=C143),B107:B112))
Average Sales	1,681,375	=AVERAGE(IF((B107:B112>C142)+(C107:C112>=C143),B107:B112))

Assembly Line Example #1 with SEQUENCE & COUNTIFS:

Post	Time to Assemble Product at Post (Seconds)
1	9
2	9.2
3	8.3
4	8.3
5	8.8
6	10
7	10.7
8	8
9	9.1
10	10.7
11	9.9
12	8.5
1	10
2	8.6
3	9.1
4	8.3
5	8.1
6	10.4
7	9.1
8	8.4
9	8.3
10	11

Goal: Count how many times a Post fell below required <10 second assembly time.

AND Logical Test: Post has to equal given post AND Seconds <10

Hurdle in Seconds:	<10

Post	Count
1	38
2	41
3	45
4	40
5	45
6	42
7	17
8	48
9	45
10	42
11	19
12	39

In Cell D8 is: =SEQUENCE(12)

SEQUENCE delivers an array of sequential number

=COUNTIFS(A2:A589,D8#,B2:B589,E5)

COUNTIFS counts with two conditions/criteria

** This is an example of where a PivotTable would be harder to create because our condition uses a comparative operator (we have a hurdle as a condition).

Assembly Line With 12 Posts:

Assembly Line Example #2 with MINIFS & MAXIFS:

**Time to Assemble Product at Post (Seconds)

Post	Seconds
1	9
2	9.2
3	8.3
4	8.3
5	8.8
6	10
7	10.7
8	8
9	9.1
10	10.3
11	9.9
12	8.5
1	10
2	8.6
3	9.1
4	8.3
5	8.1
6	10.4
7	9.1
8	8.4

Goal: Find Min Time for Each Post. And have formulas update automatically.

MINIFS MAXIFS

Post	Min	Max
1	7.8	10.5
2	7.8	10.4
3	7.5	11
4	7.8	11
5	7.3	10.5
6	7.6	10.7
7	9	12.1
8	7.5	10.4
9	7.5	11.2
10	7.2	10.3
11	9.2	11.75
12	7.5	11.1

In Cell D8 is: =SEQUENCE(12)

SEQUENCE delivers an array of sequential numbers

In Cell E8 is: =MINIFS(B4:B591,A4:A591,D8#)

MINIFS finds smallest value with one or more conditions

In Cell F8 is: =MAXIFS(B4:B591,A4:A591,D8#)

MAXIFS finds biggest value with with one or more conditions

** This is an example of where a PivotTable would be easy to create, but it would not update without a "Refresh"